
Mangle VM Programmer’s Guide
Mangle is a virtual machine for writing string transformation programs.

It runs programs composed of simple assembly language instructions that can
manipulate values, match patterns, manipulate characters, use conditional
logic, and employ subroutines.

It is intended for use in modifying existing words or generating new ones and
as such is intended to be a tool for playing with constructed languages
(conlang’s)

Core Concepts
At the core of the virtual machine is the word register w which is used for
construcing and manipulating a word. When the program is finished the value of
w is the output.

Programs that want to operate on words read from stdlin need to load them
from the in register (a readonly string register). Programs that are
generating words can skip this step and start with an empty w register.

Mangle allows for a wide range of characters (Latin–1) characters but only
works on lower case characters. The input word is automatically converted to
lower case.

Here is a simple example program:

start:
  load w, in            # Load the input word into the word buffer!
  reverse w, 0, w.len   # Reverse the entire word buffer
  halt                  # Normal end with exit 0

The result of this program is to reverse whatever word is provided on stdin.

Unicode Support
Mangle supports Latin Extended characters (U+0080-U+024F) including:
- Latin–1 Supplement: à, á, è, é, ñ, ö, ü, þ, ð
- Latin Extended-A: ā, ē, ī, ō, ū
- Latin Extended-B: Additional European characters

Character literals can be ASCII or Unicode:
- ASCII: ‘a’, ‘Z’, ‘-’
- Unicode: ‘ñ’, ‘á’, ‘é’, ‘þ’

Frequency maps support Unicode:
load tv, {a:8, e:12, i:10, o:8, u:7, á:2, é:3, í:2, ó:2, ú:1}

Note: Only precomposed characters are supported. Combining diacritics are
not handled separately.

Execution Model
Programs execute as a loop:

1. Fetch instruction at program counter (pc)
2. If the instruction is halt, stop and return state
3. Execute instruction (pure function: State → State)
4. Increment pc (jumps set pc directly)
5. Repeat

Execution halts when:
- The halt instruction is reached, or
- The program counter goes past the last instruction
- An error is set in state.error

VM Architecture

State Structure
The VM maintains the following state:

Field Type Description

chars [char()] Character list (word buffer)
vm_flag boolean Set by match/comparison instructions
pc integer Program counter

labels map Label names → instruction indices
error string or nil Error message (halts execution)
input_word string Original input word (read-only)

cursor integer() cursor register
index_registers tuple i0-i9 store integers for characters or indices
flag_registers tuple f0-f9 boolean flags

string_registers tuple s0-s9 string registers (charlists)
char_tables map ta-tz character tables
match_registers tuple m0-m9 FSM-based matchers

rand_state internal Random number generator state
call_stack list Return addresses for subroutines
max_stack_depth integer Maximum call stack depth (default: 256)
exit_code integer Exit code set by halt (0 = success, >0 = failure)

halted boolean Set when the VM has recieved a halt instruction
comment string Accumulated comment messages for explain mode
trace [any()] trace output when using –trace to the runner

Registers

General Purpose Registers (i0-i9)
Ten general-purpose registers for storing and manipulating integer values that
can either be used as characters, or as indexes into strings.

load i0, 42       # i0 = 42
load i1, i0       # i1 = i0 (copy)
load i0, 'c'      # i0 = 99 (ASCII 'c')
add i0, 10        # i0 = i0 + 10

Cursor Register (c)
The cursor register c is an index register for indexing into w. In most
respects it acts as a general purpose register except that it’s value
auto-clamps to valid bounds [0, w.len] whenever written to (whether via
load, add, inc, etc…)

load c, 0        # cursor at start
load c, w.len    # cursor at end (for appending)
load c, w.li     # cursor at last character
inc c            # move right (clamps at length)
dec c            # move left (clamps at 0)

Random Register (r)
A special register that provides random values.

Each read returns a different random value (0 to 999,999,999) and advances
the internal RNG state. Writing to the r register sets the seed. You can
either write an integer or string seed value.

# Set seed for reproducibility
load r, 42           # integer seed
load r, "planet"     # string seed
load r, w            # use current word as seed

# Get random values
load i0, r           # i0 = random value (advances RNG)
load i1, r           # i1 = different random value

Input Register (in)
A read-only register containing the original input word:

load w, in            # load input into word buffer (required!)

Word Buffer (w)
The main working string. Must be explicitly loaded from input:

load w, in            # required before any word operations

String Registers
The string registers (s0 through s9) store substrings as charlists.

Loading String Registers
load s0, w            # capture entire current word
load s0, "test"       # store literal (automatically lowercased)
load s0, ""           # clear the register
load s0, s1           # copy s1 to s0
insert s0, 1, "bar"   # insert the string "bar" at position 1 of s0

Properties
Properties provide access to lengths and last indices:

Property Description

w.len Word buffer length

w.li Word buffer last index (length - 1, clamped to 0)
s0.len - s3.len String register lengths
s0.li - s3.li String register last indices

Examples:

load i0, w.len        # i0 = word length
load c, w.li          # cursor points to last character
swap w, 0, w.li       # swap first and last characters
load i1, s0.len       # i1 = length of s0

Character Tables
Character tables (registers ta through tz) store character sets for pattern
matching and random character selection.

Loading Tables
From string (equal probability):

load tv, "aeiou"                  # vowels, equal weights
load tc, "bcdfghjklmnpqrstvwxyz"  # consonants

From frequency map (weighted probability):

load te, {a:8.2, e:12.7, i:7.0, o:7.5, u:2.8}  # English vowel frequencies
load ta, {-: 1.0, ': 2.0, a: 3.0}              # supports hyphen and apostrophe

Frequency maps support:
- Single characters (a-z)
- Digraphs (two characters: sh, th)
- Trigraphs (three characters: kth, shr)
- Special characters: hyphen (-) and apostrophe (’)

Example with special characters:

load ta, {d-d: 1, o'c: 1, o'b: 1}  # Irish name patterns

Using Tables
Insert random character:

load tv, "aeiou"
insert w, w.len, tv   # insert random vowel at end

Pattern matching:

load tv, "aeiou"
load tc, "bcdfghjklmnpqrstvwxyz"
load i0, 0
match w, i0, tc, tv   # match consonant followed by vowel
jmp_ok found

Wildcards in patterns:

match w, i0, tc, ?, tv   # consonant, any char, vowel

Match Registers
Ten match registers (m0 through m9) store FSM (Finite State Machine) based
pattern matchers. These are useful for recognizing specific sets of strings.

Building FSM Matchers
Load strings into a match register to build a finite state machine:

load m0, "bn"             # add "bn" to the matcher
load m0, "bg"             # add "bg" to the matcher
load m0, "dn"             # add "dn" to the matcher
# m0 now recognizes: "bn", "bg", "dn"

Each load adds another pattern to the FSM matcher. The matcher will recognize
any of the loaded strings.

Using FSM Matchers
Use match with a match register and a size register to capture the match length:

load m0, "ing"
load m0, "ed"
load m0, "s"

start:
  load w, in
  match w, 0, m0, i0      # try to match at position 0, store length in i0
  jmp_ok found_suffix
  halt 1

found_suffix:
  # i0 now contains the length of the matched suffix
  halt 0

The match instruction with FSM matchers:
- Sets VM flag to true if a pattern is matched
- Stores the matched length in the specified index register
- Returns 0 length if no match

Boolean Flags
Ten boolean flags (f0 through f9) store true/false values independent of
the VM flag.

Setting Flags
capturef f0           # f0 = vm_flag
setf f0, true         # explicit boolean
setf f0, false
setf f0, 42           # integer: 0 = false, non-zero = true
setf f0, c            # from register value
setf f0, w.len        # from property

Testing Flags
The testf instruction sets the VM flag based on a boolean flag:

setf f0, w.len        # f0 = true if word non-empty
testf f0              # vm_flag = true if f0 is true
jmp_ok process        # jump if f0 was true

Capturing the VM Flag
Use capturef to save the current VM flag value:

match w, 0, "th"      # sets vm_flag
capturef f0           # f0 = vm_flag
# ... other operations ...
testf f0              # restore vm_flag from f0

Boolean Logic
Boolean flags support logical operations:

setf f0, true
setf f1, false
and f0, f1            # f0 = f0 && f1
or f0, f1             # f0 = f0 || f1
not f0                # f0 = !f0
xor f0, f1            # f0 = f0 XOR f1

Copying Ranges
The copy instruction extracts a range from one register to another:

copy w, 0, 4, s0         # copy first 4 chars of word to s0
copy s0, 0, s0.len, w    # copy all of s0 to word buffer
copy w, i0, i1, s0       # copy i1 chars starting at i0 to s0

String Comparison
eq s0, s1             # compare two string registers
eq s0, "hello"        # compare with literal

Other Operations
shuffle s0            # shuffle string register
shuffle s0, 2, 3      # shuffle 3 chars starting at position 2
rotl s0, 2            # rotate left by 2
rotr s0, 1            # rotate right by 1
reverse s0            # reverse string register
reverse s0, 0, 3      # reverse first 3 chars
clear s0              # clear string register

Indexing
load_char i0, w, 0     # load char at index 0 from word buffer
load_char i0, s0, 0    # load char from string register
load_char i0, w, c     # load char at cursor position

Example: Palindrome Check
start:
  load w, in
  load s0, w          # save original word
  reverse w           # reverse the word buffer
  eq w, s0
  jmp_ok is_palindrome
  halt 1              # not a palindrome (exit code 1)

is_palindrome:
  halt 0              # it's a palindrome (exit code 0)

Instruction Reference

Documentation

debug

Syntax: debug "message"

Prints a message to stdout. Useful for development and debugging.
Does not affect VM state.

debug "starting loop"
# ... instructions ...

comment

Syntax: comment "message"

Appends a message to the VM’s internal comment field. Comments accumulate
(with newline separators) and can be displayed using --explain mode. Useful
for documenting program behavior.

comment "swapping first and last characters"
swap w, 0, w.li
comment "operation complete"

log

Syntax: log VALUE

Logs a value to stdout using IO.inspect. Useful for debugging register values
and expressions during development. Displays the value with a label describing
what was logged.

log i0                    # log integer register value
log w.len                 # log word length
log s0                    # log string register contents
log c                     # log cursor position

Control Flow

halt

Syntax: halt or halt N

Stops program execution. Optional exit code N (default 0).

halt                  # exit code 0
halt 1                # exit code 1 (failure)

jmp

Syntax: jmp LABEL

Unconditional jump to label.

jmp done
# ... skipped instructions
done:
  halt

jmp_ok

Syntax: jmp_ok LABEL

Jump if VM flag is :ok.

match w, i0, "th"
jmp_ok found          # jump if match succeeded

jmp_fail

Syntax: jmp_fail LABEL

Jump if VM flag is :fail.

match w, i0, tv
jmp_fail not_vowel    # jump if not a vowel

call

Syntax: call LABEL

Call subroutine, pushing return address.

call process_word
# ... execution continues here after ret

call_ok

Syntax: call_ok LABEL

Call subroutine if flag is :ok. If flag is :fail, execution continues to
the next instruction.

match w, c, tv         # match vowel at c
call_ok handle_vowel   # only call if match succeeded
# continues here if no match or after ret

call_fail

Syntax: call_fail LABEL

Call subroutine if flag is :fail. If flag is :ok, execution continues to
the next instruction.

match w, i0, "ing"     # try to match "ing"
call_fail try_other    # call alternative handler if no match
# continues here if match succeeded or after ret

ret

Syntax: ret

Return from subroutine.

process_word:
  load c, 0
  delete w, 0
  ret

loop

Syntax: loop COUNTER, LABEL

Counter-based loop. Decrements COUNTER and jumps to LABEL if COUNTER is not zero.
When COUNTER reaches zero, execution continues to the next instruction.

start:
  load w, in
  load i0, 3              # loop 3 times

add_vowel:
  load tv, "aeiou"
  insert w, w.len, tv     # insert random vowel at end
  loop i0, add_vowel      # decrement i0, jump if not zero
  halt

The loop instruction is useful for repeating operations a fixed number of times:

# Multiply i0 by 2, 7 times
start:
  load i0, 1
  load i1, 7              # loop counter
multiply:
  mul i0, 2
  loop i1, multiply       # loops 7 times
  # i0 is now 128
  halt

Character Operations
All character operations require explicit target and position.

insert

Syntax: insert TARGET, INDEX, SOURCE

Insert character or string at INDEX.

SOURCE can be: char literal, register (codepoint), table register (random),
string register, or string literal.

insert w, 0, 'x'          # insert 'x' at start
insert w, w.len, tv       # insert random vowel at end
insert w, c, s0           # insert string register at cursor
insert w, c, "ing"        # insert literal at cursor

delete

Syntax: delete TARGET, INDEX or delete TARGET, INDEX, COUNT

Delete one or more characters starting at INDEX.

delete w, 0               # delete first char
delete w, w.li            # delete last char
delete w, 0, 3            # delete first 3 chars
delete s0, 0, i0          # delete i0 chars from s0

swap

Syntax: swap TARGET, POS1, POS2

Swap characters at two positions. Case stays positional.

swap w, 0, w.li          # swap first and last
swap s0, i0, i1           # swap positions in i0 and i1

move

Syntax: move TARGET, FROM, TO

Move character from FROM to TO, shifting others. Case stays positional.

move w, 0, w.li          # move first char to end
move w, w.li, 0          # move last char to start

copy

Syntax: copy SOURCE, START, LEN, DEST

Copy LEN characters from SOURCE starting at START into DEST. Overwrites DEST contents.

copy w, 0, 4, s0          # first 4 chars of word into s0
copy w, 0, 2, s0          # copy first two characters of w to s0
copy s0, 0, s0.len, w     # all of s0 into word buffer
copy w, i0, i1, s1        # copy i1 chars starting at i0 into s1

replace

Syntax: replace TARGET, INDEX, SOURCE

Replace character at INDEX with SOURCE. Case stays positional.

replace w, 0, 'z'         # replace first char with 'z'
replace w, c, tv          # replace at cursor with random vowel

dup

Syntax: dup TARGET, INDEX, COUNT

Duplicate character at INDEX, inserting COUNT copies after it.

dup w, 0, 1               # "abc" → "aabc"
dup w, 0, 2               # "abc" → "aaabc"
dup s0, 0, 3              # duplicate first char 3 times

reverse

Syntax: reverse TARGET or reverse TARGET, START, LEN

Reverse characters. Case stays positional for word buffer.

reverse w                 # reverse entire word
reverse s0                # reverse string register
reverse w, 0, 3           # reverse first 3 chars

clear

Syntax: clear TARGET

Clear all characters from TARGET.

clear w                   # clear word buffer
clear s0                  # clear string register

shuffle

Syntax: shuffle TARGET or shuffle TARGET, START, LEN

Randomly shuffle characters. Uses Fisher-Yates algorithm. Case stays positional for word buffer.

shuffle w                 # shuffle word buffer
shuffle s0                # shuffle string register
shuffle s0, 2, 3          # shuffle 3 chars starting at position 2

rotl

Syntax: rotl TARGET, COUNT

Rotate characters left. Case stays positional for word buffer.

rotl w, 1                 # "hello" → "elloh"
rotl s0, 2                # rotate s0 left by 2

rotr

Syntax: rotr TARGET, COUNT

Rotate characters right. Case stays positional for word buffer.

rotr w, 1                 # "hello" → "Ohell"
rotr s0, 2                # rotate s0 right by 2

load_char

Syntax: load_char DEST, SOURCE, INDEX

Load character at INDEX from SOURCE into DEST register.

load_char i0, w, 0         # load first char from word buffer
load_char i0, s0, 0        # load first char from s0
load_char i0, w, c         # load char at cursor

Pattern Matching

match

Syntax: match TARGET, POS, PATTERN... or match TARGET, POS, MATCH_REG,
SIZE_REG

Match pattern at POS in TARGET. Sets VM flag.

Pattern elements:
- String literals: "hello"
- Table registers: tv, tc
- Register values (codepoints): i0
- String registers: s0
- Wildcard: ?
- FSM match registers: m0-m9

# Literal match
match w, c, "th"

# Table sequence
match w, c, tc, tv        # consonant then vowel

# With wildcard
match w, c, tc, ?, tv     # consonant, any, vowel

# Register value
load i0, 97               # 'a'
match w, i0, i0

# String register
match w, c, s0

FSM Matcher syntax: match TARGET, POS, MATCH_REG, SIZE_REG

When using an FSM match register, a size register captures the matched length:

# Build an FSM matcher
load m0, "ing"
load m0, "ed"
load m0, "tion"

# Match and capture length
match w, 0, m0, i0        # match at position 0, length stored in i0
jmp_ok found
halt 1

found:
  # i0 contains length of matched pattern (e.g., 3 for "ing")
  halt 0

Register Operations

load

Syntax: load DEST, SOURCE

Load value into register.

load i0, 42               # integer
load i1, i0               # copy register
load i0, w.len            # property
load i0, @char            # special value
load i0, rl               # match length
load w, in                # input to word buffer
load tv, "aeiou"          # table from string
load tv, {a:8, e:12}      # table from frequency map
load r, 42               # set random seed
load s0, @word            # capture word
load s0, "test"           # literal string

inc

Syntax: inc REG

Increment register by 1.

inc i0
inc c                    # cursor moves right (clamps)

dec

Syntax: dec REG

Decrement register by 1.

dec i0
dec c                    # cursor moves left (clamps at 0)

add

Syntax: add REG, VALUE

Add VALUE to register.

add i0, 10
add i0, i1
add i0, w.len

sub

Syntax: sub REG, VALUE

Subtract VALUE from register.

sub i0, 3
sub i0, i1

mul

Syntax: mul REG, VALUE

Multiply register by VALUE.

mul i0, 2

div

Syntax: div REG, VALUE

Integer divide register by VALUE. Errors if VALUE is 0.

div i0, 2

mod

Syntax: mod REG, VALUE

Compute register modulo VALUE. Errors if VALUE is 0.

load i0, r
mod i0, 10                # i0 = random 0-9

Comparison Operations
All set VM flag to :ok if true, :fail if false.

eq

Test equality.

eq i0, 5
eq i0, i1
eq s0, s1                 # string comparison
eq s0, "test"             # string vs literal

neq

Test inequality.

neq i0, 0

lt

Test less than.

lt c, w.len              # cursor not at end?

lte

Test less than or equal.

lte i0, 10

gt

Test greater than.

gt w.len, 3               # word longer than 3?

gte

Test greater than or equal.

gte c, w.len             # cursor at or past end?

Boolean Flag Operations

setf

Syntax: setf FLAG, VALUE

Set a boolean flag based on a truthy value. Integers are converted: 0 is false, non-zero is true.

setf f0, true
setf f0, false
setf f0, w.len            # true if non-empty
setf f0, i0               # true if i0 != 0

testf

Syntax: testf FLAG

Set VM flag based on boolean flag. Sets VM flag to true if the flag is true, false otherwise.

testf f0
jmp_ok process

capturef

Syntax: capturef FLAG

Capture the current VM flag value into a boolean flag register.

match w, 0, "th"          # sets vm_flag
capturef f0               # save the result to f0
# ... other operations that might change vm_flag ...
testf f0                  # restore from f0
jmp_ok handle_th

and

Syntax: and FLAG1, FLAG2

Logical AND of two flag registers. Result stored in FLAG1.

setf f0, true
setf f1, false
and f0, f1                # f0 = f0 && f1 = false

or

Syntax: or FLAG1, FLAG2

Logical OR of two flag registers. Result stored in FLAG1.

setf f0, false
setf f1, true
or f0, f1                 # f0 = f0 || f1 = true

not

Syntax: not FLAG

Logical NOT of a flag register. Inverts the flag value.

setf f0, true
not f0                    # f0 = !f0 = false

xor

Syntax: xor FLAG1, FLAG2

Logical XOR of two flag registers. Result stored in FLAG1.

setf f0, true
setf f1, true
xor f0, f1                # f0 = (f0 && !f1) || (!f0 && f1) = false

Program Structure

Syntax
# This is a comment
label_name:
  instruction arg1, arg2
  instruction arg1
  instruction

Required: Load Input
Every program that wants to work on input words should start by loading the
input word (in in) into w.

start:
  load w, in              # Required!
  # ... rest of program
  halt

Labels
Labels are defined with a trailing colon:

start:
  jmp process

done:
  halt

process:
  # do work
  jmp done

Arguments
Type Syntax Examples

Integer digits 42, 0

Register identifier i0, c, w, s0
Property reg.prop w.len, s0.li
Label identifier start, loop

String "..." "hello"

Character 'c' 'a', 'Z'
Frequency map {c:n, ...} {a:8.2, e:12.7}

Wildcard ? Used in match
Boolean true/false Used with set

Programming Patterns

Swap First and Last Characters
start:
  load w, in
  swap w, 0, w.li
  halt

Input “Planet” → Output “Tlanep”

Delete First Character
start:
  load w, in
  delete w, 0
  halt

Strip Consonant Cluster from Start
start:
  load w, in
  load tc, "bcdfghjklmnpqrstvwxyz"

loop:
  load i0, 0
  match w, i0, tc
  jmp_fail done
  delete w, 0
  jmp loop

done:
  halt

Reverse Word
start:
  load w, in
  reverse w
  halt

Input “hello” → Output “olleh”

Insert Random Vowel at End
start:
  load w, in
  load tv, "aeiou"
  insert w, w.len, tv
  halt

Loop Through Characters
start:
  load w, in
  load c, 0

loop:
  gte c, w.len
  jmp_ok done

  load_char i0, w, c      # get current char
  # ... process character

  inc c
  jmp loop

done:
  halt

Using Subroutines
start:
  load w, in
  call process_start
  call process_end
  halt

process_start:
  delete w, 0
  ret

process_end:
  delete w, w.li
  ret

Random with Reproducible Seed
start:
  load w, in
  load r, w               # seed with input word
  load tv, "aeiou"
  insert w, w.len, tv     # deterministic "random" based on input
  halt

Running Programs

Command Line
# Basic usage
echo "hello" | mangle program.mangle

# Multiple words
cat words.txt | mangle program.mangle

# With execution trace
echo "hello" | mangle --trace program.mangle

# With explain mode (shows comment messages)
echo "hello" | mangle --explain program.mangle

# Skip empty lines from input
cat words.txt | mangle --ignore-empty program.mangle

# Quiet mode (output words only, no T/F prefix)
echo "hello" | mangle --quiet program.mangle

# Comparison mode (shows original and transformed word)
echo "hello" | mangle --comparison program.mangle

# Set maximum call stack depth
echo "hello" | mangle --stack-depth 100 program.mangle

# Run N times with empty input (useful for generators)
mangle --n 5 generator.mangle

CLI Options
Option Description

--trace Print step-by-step execution trace for each word
--explain Print results with accumulated comment messages

--ignore-empty Skip empty lines from stdin
--quiet Output only transformed words (no T/F success prefix)
--stack-depth N Set maximum call stack depth (default: 256)

--n N Run program N times with empty input (bypasses stdin)
-i0 through -i9 Pre-initialize integer register (e.g., -i0 42)
-s0 through -s9 Pre-initialize string register (e.g., -s0 "hello")
-f0 through -f9 Pre-initialize flag register (e.g., -f0 true, -f1 f)

Register Initialization
Pre-initialize VM registers before program execution:

# Initialize i0 to 5 and s1 to "prefix"
echo "test" | mangle -i0 5 -s1 "prefix" program.mangle

# Initialize flag f0 to true (accepts: true, false, t, f)
echo "test" | mangle -f0 t program.mangle

# Multiple initializations
mangle -i0 10 -i1 20 -s0 "seed" -f0 true --n 100 generator.mangle

Output Format
By default, output includes a success indicator:

T transformed_word   # T = success (exit code 0)
F transformed_word   # F = failure (exit code > 0)

With --quiet, only the transformed word is output.

Programmatic Usage (Elixir)
# Parse a program
{:ok, %{program: program}} = Mangle.Parser.parse(source)

# Process a single word
state = Mangle.State.from_word("hello")
result = Mangle.Runner.run(program, state)
output = Mangle.State.to_word(result)

# Process multiple words
outputs = Mangle.Harness.run(program, ["hello", "World"])

Debugging

Execution Trace
Use --trace for step-by-step execution:

--- Hello → T Oellh ---
PC | F | Instruction        | Word    | Changes
---+---+--------------------+---------+--------
00 | T | load w, in         | "hello" | word loaded
01 | T | swap w, 0, w.li   | "Oellh" | word changed
02 | T | halt               | "Oellh" |

Explain Mode
Use --explain to see comment messages accumulated during execution:

echo "hello" | mangle --explain program.mangle

Output:

T Oellh
# swapping first and last characters
# operation complete

This is useful for documenting complex program behavior and understanding what
transformations were applied.

Example program with comments:

start:
  load w, in
  comment "swapping first and last characters"
  swap w, 0, w.li
  comment "operation complete"
  halt

Error Handling
Errors halt execution immediately. Common errors:

Error Cause

swap: cannot swap in empty word Operating on empty word

swap: position N is out of bounds Index >= length
uninitialized table register tX Using table before loading
div: division by zero Dividing by 0

call: maximum call depth exceeded Too many nested calls

Defensive Programming
start:
  load w, in
  eq w.len, 0
  jmp_ok done             # Skip empty words

  # Safe to operate
  delete w, 0

done:
  halt

Quick Reference

Registers
i0-i9: General purpose integer registers
c: Cursor (auto-clamps to [0, w.len])
r: Random (read advances RNG, write seeds)
in: Input word (read-only)
w: Word buffer
ta-tz: Character tables
s0-s9: String registers
f0-f9: Boolean flags
m0-m9: FSM match registers

Properties
w.len, w.li: Word buffer length/last index
s0.len, s0.li, etc.: String register length/last index
in.len, in.li: Input word length/last index

Essential Instructions
load w, in                    # Load input (required!)
load REG, VALUE               # Load value
inc/dec REG                   # Increment/decrement
add/sub/mul/div/mod REG, VAL  # Arithmetic
eq/neq/lt/lte/gt/gte A, B     # Compare (sets vm_flag)
jmp/jmp_ok/jmp_fail LABEL     # Jump
call/call_ok/call_fail/ret    # Subroutines
loop COUNTER, LABEL           # Counter-based loop
match TARGET, POS, PATTERN    # Pattern match (sets vm_flag)
match TARGET, POS, mN, iN     # FSM match (sets vm_flag, stores length)
insert TARGET, INDEX, SOURCE  # Insert
delete TARGET, INDEX[, COUNT] # Delete
swap TARGET, POS1, POS2       # Swap
move TARGET, FROM, TO         # Move
copy SRC, START, LEN, DEST    # Copy range
replace TARGET, INDEX, SOURCE # Replace
dup TARGET, INDEX, COUNT      # Duplicate
reverse TARGET[, START, LEN]  # Reverse
shuffle TARGET[, START, LEN]  # Shuffle
rotl/rotr TARGET, COUNT       # Rotate
clear TARGET                  # Clear
load_char DEST, SRC, INDEX    # Load char at index
setf FLAG, VALUE              # Set boolean flag
testf FLAG                    # Test boolean flag (sets vm_flag)
capturef FLAG                 # Capture vm_flag to boolean flag
and/or/xor FLAG1, FLAG2       # Boolean logic
not FLAG                      # Boolean negation
debug "message"               # Print debug output
comment "message"             # Add explanation comment
log VALUE                     # Log value with IO.inspect
halt                          # Stop (exit code 0)
halt N                        # Stop with exit code N

